```
"GIVEN:  A Steam power cycle with reheat as shown,"

    s$ = 'steam_iapws'
    P[1] = 1600[psia]
    T[1] = 1100[F]
$IFNOT Parametric Study or MinMax
    P[2] = 300[psia] "<-- initial guess value from exam"
    P[4] = 125 [psia] "<-- initial guess value from exam"
$ENDIF
    P[3] = P[2]
    P[5] = P[4]
    T[3] = 900 [F]
    T[5] = 750[F]
    P[6] = 1 [psia]
    x[7] = 0 [dim]
    P[7] = P[6]
    T[7] = T_sat(s$, P=P[7])
    P[8] = P[1]
    eta_turbine = 0.85
    eta_pump = 0.90
    W_dot_net = 200000*convert(hp, Btu/hr)
    T_carnot_H = 1100 [F]
    T_carnot_L = 162 [F]

"FIND:"

"SOLUTION:"
"!Build the property table first"
"State 1"
    h[1] = enthalpy(s$,P=P[1],T=T[1]);    s[1] = entropy(s$,P=P[1],T=T[1]);        x[1] = quality(s$,P=P[1],T=T[1]);        v[1] =
volume(s$,P=P[1],T=T[1])
"State 2s"
    eta_turbine = (h[1] - h[2])/(h[1] - h_2s)
    h_2s = enthalpy(s$,P=P[2],s=s_2s);   s_2s = s[1]
"State 2"
    s[2] = entropy(s$,P=P[2],h=h[2]);       T[2] = temperature(s$,P=P[2],h=h[2]);   x[2] = quality(s$,P=P[2],h=h[2]);        v[2] =
volume(s$,P=P[2],h=h[2])
"State 3"
    h[3] = enthalpy(s$,P=P[3],T=T[3]);      s[3] = entropy(s$,P=P[3],T=T[3]);        x[3] = quality(s$,P=P[3],T=T[3]);        v[3] =
volume(s$,P=P[3],T=T[3])
"State 4s"
    eta_turbine = (h[3] - h[4])/(h[3] - h_4s)
    h_4s = enthalpy(s$,P=P[4],s=s_4s);    s_4s = s[3]
"State 4"
    s[4] = entropy(s$,P=P[4],h=h[4]);       T[4] = temperature(s$,P=P[4],h=h[4]);   x[4] = quality(s$,P=P[4],h=h[4]);        v[4] =
volume(s$,P=P[4],h=h[4])
"State 5"
    h[5] = enthalpy(s$,P=P[5],T=T[5]);      s[5] = entropy(s$,P=P[5],T=T[5]);        x[5] = quality(s$,P=P[5],T=T[5]);        v[5] =
volume(s$,P=P[5],T=T[5])
"State 6s"
    eta_turbine = (h[5] - h[6])/(h[5] - h_6s)
    h_6s = enthalpy(s$,P=P[6],s=s_6s);    s_6s = s[5]
"State 6"
    s[6] = entropy(s$,P=P[6],h=h[6]);       T[6] = temperature(s$,P=P[6],h=h[6]);   x[6] = quality(s$,P=P[6],h=h[6]);        v[6] =
volume(s$,P=P[6],h=h[6])
"State 7"
    h[7] = enthalpy(s$,x=x[7],T=T[7]);      s[7] = entropy(s$,x=x[7],T=T[7]);        v[7] = volume(s$,T=T[7],x=x[7])
"State 8_s"
    eta_pump = (h_8s - h[7])/(h[8] - h[7])
    h_8s = enthalpy(s$,P=P[8],s=s_8s);     s_8s = s[7]
"State 8"
    s[8] = entropy(s$,P=P[8],h=h[8]);         T[8] = temperature(s$,P=P[8],h=h[8]);  x[8] = quality(s$,P=P[8],h=h[8]);        v[8] =
volume(s$,P=P[8],h=h[8])
```

```
"!Thermodynamics!"
"The mass flow rate of the steam is found from the net power deliverd by the cycle,"
    W_dot_net = W_dot_t - W_dot_p
"The total power delivery of the turbines is,"
    W_dot_t = W_dot_turbine1 + W_dot_turbine2 + W_dot_turbine3
"The first turbine power delivery is,"
    W_dot_turbine1 = m_dot*(h[1] - h[2])
"The second turbine power delivery is,"
    W_dot_turbine2 = m_dot*(h[3] - h[4])
"The third turbine power delivery is,"
    W_dot_turbine3 = m_dot*(h[5] - h[6])
"The pump power requirement is,"
    W_dot_p = m_dot*(h[8] - h[7])
"The heat transfer rate in the boiler is,"
    Q_dot_in = m_dot*((h[1] - h[8]) + (h[3] - h[2]) + (h[5] - h[4]))
"The condenser heat transfer rate is,"
    Q_dot_out = m_dot*(h[6] - h[7])
"The thermal efficiency of the cycle is,"
    eta_th = W_dot_net/Q_dot_in
"The heat rate of the cycle is,"
    HR = Q_dot_in/(W_dot_net*convert(Btu/hr,kW))

eta_carnot = 1-(converttemp(F,R, T_carnot_L)/converttemp(F,R, T_carnot_H))
```

SOLUTION

Unit Settings: Eng F psia mass deg

η_{carnot} = 0.6014
η_{pump} = 0.9
η_{th} = 0.38622
$\eta_{turbine}$ = 0.85
HR = 8835 [Btu/kW-hr]
h_{2s} = 1320 [Btu/lb$_m$]
h_{4s} = 1357 [Btu/lb$_m$]
h_{6s} = 1005 [Btu/lb$_m$]
h_{8s} = 74.49 [Btu/lbm]
$\dot{m}$ = 812411 [lbm/hr]
$\dot{Q}_{in}$ = 1.318E+09 [Btu/hr]
$\dot{Q}_{out}$ = 8.087E+08 [Btu/hr]
s$ = 'steam_iapws'
s_{2s} = 1.632 [Btu/lb$_m$-R]
s_{4s} = 1.759 [Btu/lb$_m$-R]
s_{6s} = 1.8 [Btu/lb$_m$-R]
s_{8s} = 0.1326 [Btu/lbm-R]
$T_{carnot,H}$ = 1100 [F]
$T_{carnot,L}$ = 162 [F]
$\dot{W}_{net}$ = 5.089E+08 [Btu/hr]
$\dot{W}_p$ = 4.300E+06 [Btu/hr]
$\dot{W}_t$ = 5.132E+08 [Btu/hr]
$\dot{W}_{turbine1}$ = 1.574E+08 [Btu/hr]
$\dot{W}_{turbine2}$ = 8.096E+07 [Btu/hr]
$\dot{W}_{turbine3}$ = 2.748E+08 [Btu/hr]

No unit problems were detected.

KEY VARIABLES

$\dot{m}$ = 812411 [lbm/hr]	*(b) mass flow rate of steam in the cycle*
$\dot{W}_p$ = 4.300E+06 [Btu/hr]	*(d) Power input required by the pump*
$\dot{Q}_{out}$ = 8.087E+08 [Btu/hr]	*(e) Heat transfer rate at the condenser (a loss)*
$\dot{Q}_{in}$ = 1.318E+09 [Btu/hr]	*(f) Heat transfer rate at the boiler (input)*
η_{th} = 0.38622	*(g) Thermal efficiency of the cycle*
HR = 8835 [Btu/kW-hr]	*(h) Heat rate of the cycle*

Arrays Table: Main

	P_i [psia]	T_i [F]	h_i [Btu/lb$_m$]	s_i [Btu/lb$_m$-R]	x_i [dim]	v_i [ft^3/lbm]
1	1600	1100	1548	1.632	100	0.549
2	300	672.9	1354	1.663	100	2.168

Arrays Table: Main

	P_i [psia]	T_i [F]	h_i [Btu/lbm]	s_i [Btu/lbm-R]	x_i [dim]	v_i [ft3/lbm]
3	300	900	1474	1.759	100	2.653
4	125	692.3	1374	1.775	100	5.414
5	125	750	1403	1.8	100	5.699
6	1	101.7	1065	1.906	0.9611	320.5
7	1	101.7	69.72	0.1326	0	0.01614
8	1600	102.8	75.01	0.1336	-100	0.01606

REFLECTION:

Thermal efficiency reaches maximum of 38.2% at ~175 psia reheat pressure.
This is slightly higher than the cycle w/o reheat which had an efficiency of 36.4%.

Mass flow rate is fairly constant, increasing slightly at reheat pressure increases.
The mass flow rate is somewhat less than the cycle w/o reheat.

Boiler duty and condenser duty have similar behavior to mass flow rate.

The ratio of HPT to LPT power decreases substantially as reheat pressure increases.

As expected, the quality at the turbine exit is somewhat higher than the cycle w/o reheat.